In[ ]:=
```
(* P(G1) and P(G1a),  under the IM model for 3 sequences (a1, a2, b) for heuristic species delimitation.
   The notation is from Kornai et al. (2024).
    PG1a works if M>0 only.  PG1aInt works if M is specified with high precision,
   such as M = 10`100.
 *)
(* theta = 2 is fixed so that 2*tau/theta is branch length in coalescent units *)
θ = 2;
Q[M_] := Block[{θ1, θ2, M12, M21, w12, w21},
   θ1 = θ; θ2 = θ; M12 = M; M21 = M;
   c1 = 2 / θ1; c2 = 2 / θ2; w12 = 4 M12 / θ2; w21 = 4 M21 / θ1;

    {{-3 (c1 + w21), w21, w21, 0, w21, 0, 0, 0, c1, c1, c1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0},
     {w12, -(c1 + w12 + 2 w21), 0, w21, 0, w21, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, c1, 0},
     {w12, 0, -(c1 + w12 + 2 w21), w21, 0, 0, w21, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, c1, 0, 0},
     {0, w12, w12, -(c2 + 2 w12 + w21), 0, 0, 0, w21, 0, 0, 0, 0, 0, 0, c2, 0, 0, 0, 0, 0, 0},
     {w12, 0, 0, 0, -(c1 + w12 + 2 w21), w21, w21, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, c1, 0, 0, 0},
     {0, w12, 0, 0, w12, -(c2 + 2 w12 + w21), 0, w21, 0, 0, 0, 0, 0, 0, 0, c2, 0, 0, 0, 0, 0},
     {0, 0, w12, 0, w12, 0, -(c2 + 2 w12 + w21), w21, 0, 0, 0, 0, 0, 0, 0, c2, 0, 0, 0, 0},
     {0, 0, 0, w12, 0, w12, w12, -3 (c2 + w12), 0, 0, 0, c2, c2, c2, 0, 0, 0, 0, 0, 0, 0},
     {0, 0, 0, 0, 0, 0, 0, 0, -2 w21 - c1, 0, 0, 0, 0, 0, w21, 0, 0, w21, 0, 0, c1},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, -2 w21 - c1, 0, 0, 0, 0, 0, w21, 0, 0, w21, 0, c1},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -2 w21 - c1, 0, 0, 0, 0, 0, w21, 0, 0, w21, c1},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -2 w12 - c2, 0, 0, w12, 0, 0, w12, 0, 0, c2},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -2 w12 - c2, 0, 0, w12, 0, 0, w12, 0, c2},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -2 w12 - c2, 0, 0, w12, 0, 0, w12, c2},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, w12, 0, 0, w21, 0, 0, - w12 - w21, 0, 0, 0, 0, 0, 0},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, w12, 0, 0, w21, 0, 0, - w12 - w21, 0, 0, 0, 0, 0},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, w12, 0, 0, w21, 0, 0, - w12 - w21, 0, 0, 0, 0},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, w12, 0, 0, w21, 0, 0, 0, 0, 0, - w12 - w21, 0, 0, 0},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, w12, 0, 0, w21, 0, 0, 0, 0, 0, - w12 - w21, 0, 0},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, w12, 0, 0, w21, 0, 0, 0, 0, 0, - w12 - w21, 0},
     {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}}
  ];
```

In[ ]:=
```
PG1a[τ_, M_] := Block[{ Esys, U, Evalues, P, θ1, θ2},
   θ1 = θ; θ2 = θ;
   Esys = Eigensystem[Q[M]];
   Evalues = Esys[[1]];
   If[Abs[Evalues[[21]] ] > 10^(-20), Print["eigenvalue 21 is not 0?"]];
   Evalues[[21]] = -1;  (* eigenvalues are ordered increasingly, last one is 0. *)
   Evalues = (Exp[Evalues * τ] - 1) / Evalues;
   Evalues[[21]] = τ;
   U = Transpose[Esys[[2]]];
   P = Chop[U . DiagonalMatrix[Evalues] . Inverse[U]];
   (P[[2, 1]] + P [[2, 2]]) * 2 / θ1 + (P[[2, 7]] + P [[2, 8]]) * 2 / θ2
  ];
PG1b[τ_, M_] := Block[{P},
   P = MatrixExp[Q[M] * τ];
   (P[[2, 1]] + P[[2, 2]] + P[[2, 3]] + P[[2, 4]] + P[[2, 5]] + P[[2, 6]] + P[[2, 7]] + P[[2, 8]]) / 3
  ];
PG1[τ_, M_] := PG1a[τ, M] + PG1b[τ, M];
```

In[ ]:=
```
b = 10; Mmin = 0.001; Mmax = 10; τmin = 0.01; τmax = 10;
newStyle[x_] := x /. l_Line :> Sequence[Opacity[.4], Thick, Red, l]
ContourPlot[(PG1[b^τ, b^M] - 1/3) * 3/2,
    {τ, Log[b, τmin], Log[b, τmax]}, {M, Log[b, Mmin], Log[b, Mmax]},
    Contours → {0.025, 0.05, 0.1, 0.15, 0.2, 0.25, 0.3, 0.4, 0.5, 0.6, 0.7, 0.8, 0.9, 0.95},
    (* ContourStyle→{{Red, Thin}}, *)
    BaseStyle → {FontFamily → "Arial", FontSize → 9}, PlotPoints → 30, ContourLabels → All,
    ColorFunction → (ColorData[{"BeachColors", "Reverse"}]), ContourStyle → Thin, AspectRatio → 1,
    (* Frame→False, *)
    FrameTicks → {Table[{τ, ToString[Round[b^τ, τmin]]}, {τ, Log[b, τmin], Log[b, τmax]}],
      Table[{M, ToString[Round[b^M, Mmin]]}, {M, Log[b, Mmin], Log[b, Mmax]}]}
  ] /. Tooltip[x_, 0.2] :> Tooltip[newStyle[x], 0.2] /.
 Tooltip[x_, 0.7] :> Tooltip[newStyle[x], 0.7]

ContourPlot[PG1a[b^τ, b^M], {τ, Log[b, τmin], Log[b, τmax]}, {M, Log[b, Mmin], Log[b, Mmax]},
    Contours → {0.025, 0.05, 0.1, 0.15, 0.2, 0.25, 0.3, 0.4, 0.5, 0.6, 0.7, 0.8, 0.9, 0.95},
    BaseStyle → {FontFamily → "Arial", FontSize → 9}, PlotPoints → 30, ContourLabels → All,
    ColorFunction → (ColorData[{"BeachColors", "Reverse"}]), ContourStyle → Thin, AspectRatio → 1,
    (* Frame→False, *)
    FrameTicks → {Table[{τ, ToString[Round[b^τ, τmin]]}, {τ, Log[b, τmin], Log[b, τmax]}],
      Table[{M, ToString[Round[b^M, Mmin]]}, {M, Log[b, Mmin], Log[b, Mmax]}]}
  ] /. Tooltip[x_, 0.2] :> Tooltip[newStyle[x], 0.2] /.
 Tooltip[x_, 0.7] :> Tooltip[newStyle[x], 0.7]
```

Out[ ]=



Out[ ]=